

*response var.*

## STAT 5200 Handout #9: Factorial Design (Ch. 8)

**Example:** An experiment is carried out to evaluate the effects of two factors on the percentage **Yield** of an industrial process. The factors are:

- The amount of **Base** added. This factor has two levels, coded 1 and 2.
- The amount of **Alcohol** added. This factor has three levels, coded 1, 2, and 3.

Four batches with each combination of levels of Alcohol and Base are made in a completely random order. From each batch, a single measurement of the percent yield of the process is made.

```
/* Enter data */
data chemical;
  input Base Alcohol Yield @@;
  label Yield = "Percent Yield";
  cards;
    1 1 91.3 1 1 89.9 1 1 90.7 1 1 91.4 1 2 89.3
    1 2 88.1 1 2 90.4 1 2 91.4
    1 3 89.5 1 3 87.6 1 3 88.3 1 3 90.3 2 1 87.3
    2 1 89.4 2 1 91.5 2 1 88.3
    2 2 92.3 2 2 91.5 2 2 90.6 2 2 94.7 2 3 93.1
    2 3 90.7 2 3 91.5 2 3 89.8
  ;
run;
```

```
/* Fit two-way ANOVA model (with interaction)
and look at basic post hoc means comparisons */
proc glm data=chemical plots=diagnostic;
  class Base Alcohol;
  model Yield = Alcohol|Base;
  /* = Alcohol Base Alcohol*Base */
  lsmeans Base*Alcohol / pdiff adjust=Tukey lines;
  title1 'Two-way ANOVA';
run;
```

*first order puts on horiz. axis in interaction plot*


*tells to include interaction*

Two-way ANOVA		
Class	Levels	Values
Base	2	1 2
Alcohol	3	1 2 3


Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	34.47208333	6.89441667	3.38	0.0252
Error	18	36.75750000	2.04208333		
Corrected Total	23	71.22958333			

R-Square	Coeff Var	Root MSE	Yield Mean
0.483957	1.581279	1.429015	90.37083

Source	DF	Type III SS	Mean Square	F Value	Pr > F
Alcohol	2	5.39583333	2.69791667	1.32	0.2915
Base	1	6.51041667	6.51041667	3.19	0.0910
Base*Alcohol	2	22.56583333	11.28291667	5.53	0.0135



Alcohol level 1 is driving the interaction; something different there causes interaction to be signif.





How does the effect of Base depend on Alcohol?

- Increase (from Base 1 to Base 2) when Alc = 2, 3
- Decrease when Alc = 1

Least Squares Means  
Adjustment for Multiple Comparisons: Tukey

Base	Alcohol	Yield LSMEAN	LSMEAN Number
1	1	90.8250000	1
1	2	89.8000000	2
1	3	88.9250000	3
2	1	89.1250000	4
2	2	92.2750000	5
2	3	91.2750000	6

Least Squares Means for effect Base*Alcohol Pr >  t  for H0: LSMean(i)=LSMean(j)						
i/j	1	2	3	4	5	6
1		0.9071	0.4441	0.5593	0.7066	0.9974
2	0.9071		0.9499	0.9833	0.1915	0.6923
3	0.4441	0.9499		0.9999	0.0380	0.2344
4	0.5593	0.9833	0.9999		0.0563	0.3173
5	0.7066	0.1915	0.0380	0.0563		0.9154
6	0.9974	0.6923	0.2344	0.3173	0.9154	



### Tukey Comparison Lines for Least Squares Means of Base\*Alcohol


LS-means with the same letter are not significantly different.

		Yield LSMEAN	Base	Alcohol	LSMEAN Number
	A	92.275	2	2	5
	A				
B	A	91.275	2	3	6
B	A				
B	A	90.825	1	1	1
B	A				
B	A	89.800	1	2	2
B	A				
B	A	89.125	2	1	4
B					
B		88.925	1	3	3

```

/* Sample Box-Cox code */
proc transreg data=chemical;
  model boxcox(Yield / lambda = -10 to 15 by 1)
    = class( Alcohol | Base );
  title1 'Sample Box-Cox output';
run;

```



```

/* Look at other interaction plot, and do some contrasts */
proc glm data=chemical;
  class Alcohol Base; /* Notice order change here */
  model Yield = Alcohol|Base;
  contrast 'Test Alcohol, by hand' Alcohol 1 -1 0, "and"
    Alcohol 1 0 -1;
  contrast 'Test Base, by hand' Base 1 -1;
  title1 'Two-way ANOVA';
  title2 '(and testing main effects by hand with contrasts)';
run;

```


<b>Two-way ANOVA</b>					
<i>(and testing main effects by hand with contrasts)</i>					
<b>Source</b>	<b>DF</b>	<b>Sum of Squares</b>	<b>Mean Square</b>	<b>F Value</b>	<b>Pr &gt; F</b>
<b>Model</b>	5	34.47208333	6.89441667	3.38	0.0252
<b>Error</b>	18	36.75750000	2.04208333		
<b>Corrected Total</b>	23	71.22958333			

<b>R-Square</b>	<b>Coeff Var</b>	<b>Root MSE</b>	<b>Yield Mean</b>
0.483957	1.581279	1.429015	90.37083

Source	DF	Type III SS	Mean Square	F Value	Pr > F
Alcohol	2	5.39583333	2.69791667	1.32	0.2915
Base	1	6.51041667	6.51041667	3.19	0.0910
Alcohol*Base	2	22.56583333	11.28291667	5.53	0.0135

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
Test Alcohol, by hand	2	5.39583333	2.69791667	1.32	0.2915
Test Base, by hand	1	6.51041667	6.51041667	3.19	0.0910



(Here is the “original” Interaction plot from page 3, for comparison.)

